
Change 1 
	Function name
	f_EUTRA_UE_Detach_SwitchOffUe_Idle

	Reason for change
	1. Function name is confusing

2. parameter p_RRC_TI is not needed
3. optional parameters of f_EUTRA_Activate_SRB2_DRB_SendRrcMsg and f_EUTRA_Activate_SRB2_DRB_ReceiveRrcAndNasMsg can be left out

	Summary of change
	1. function renamed

2. p_RRC_TI removed
3. optional parameters removed from call to f_EUTRA_Activate_SRB2_DRB_SendRrcMsg and f_EUTRA_Activate_SRB2_DRB_ReceiveRrcAndNasMsg

	Note
	function has been introduced as documented in MCC160 comments to R5s150190

	TTCN module
	EUTRA_NASSteps


Before

	  function f_EUTRA_UE_Detach_SwitchOffUe_Idle(EUTRA_CellId_Type p_CellId,

                                              template (present) NAS_MSG_Indication_Type p_ExpectedDetachRequest,

                                              template (present) NAS_MSG_Indication_Type p_ExpectedServiceRequest,

                                              RRC_TransactionIdentifier p_RRC_TI := tsc_RRC_TI_Def,

                                              integer p_SelectedPlmnId := 1) runs on EUTRA_PTC

  { // @sic R5s150190r1 sic@

    var NAS_MSG_Indication_Type v_NasInd;    

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(p_CellId,

                                                 p_RRC_TI, ?, // @sic R5s100766 sic@

                                                 (p_ExpectedDetachRequest, p_ExpectedServiceRequest), // @sic R5-131868 sic@

                                                 p_SelectedPlmnId); // @sic R5s110468 sic@

    if (ischosen(v_NasInd.Pdu.Msg.sERVICE_REQUEST)) { // @sic R5-131868 sic@

      f_EUTRA_Activate_SRB2_DRB_SendRrcMsg(p_CellId,

                                           p_RRC_TI,

                                           v_NasInd.SecurityProtection.NasCount,
                                           0,
                                           0,
                                           omit,
                                           omit,
                                           omit,
                                           omit,
                                           omit,
                                           true,
                                           omit);

      f_EUTRA_Activate_SRB2_DRB_ReceiveRrcAndNasMsg(p_CellId,

                                                    p_RRC_TI,
                                                    0,
                                                    0);

      // DRB configured, so when the UE has finished Deregistration on IMS, it will then send the Detach..

      SRB.receive(car_SRB2_NasPdu_IND(p_CellId, p_ExpectedDetachRequest));

    }
  }


After
	  function f_EUTRA_UE_DetachMO_Idle(EUTRA_CellId_Type p_CellId,

                                    template (present) NAS_MSG_Indication_Type p_ExpectedDetachRequest,

                                    template (present) NAS_MSG_Indication_Type p_ExpectedServiceRequest,

                                    integer p_SelectedPlmnId := 1) runs on EUTRA_PTC

  { // @sic R5s150190r1 sic@

    var NAS_MSG_Indication_Type v_NasInd;

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(p_CellId,

                                               v_RRC_TI, 

                                               ?, // @sic R5s100766 sic@

                                               (p_ExpectedDetachRequest, p_ExpectedServiceRequest), // @sic R5-131868 sic@

                                               p_SelectedPlmnId); // @sic R5s110468 sic@

    if (ischosen(v_NasInd.Pdu.Msg.sERVICE_REQUEST)) { // @sic R5-131868 sic@

      f_EUTRA_Activate_SRB2_DRB_SendRrcMsg(p_CellId,

                                           v_RRC_TI,

                                           v_NasInd.SecurityProtection.NasCount);

      f_EUTRA_Activate_SRB2_DRB_ReceiveRrcAndNasMsg(p_CellId, v_RRC_TI);

      // DRB configured, so when the UE has finished Deregistration on IMS, it will then send the Detach..

      SRB.receive(car_SRB2_NasPdu_IND(p_CellId, p_ExpectedDetachRequest));

    }        

  }


Change 2 
	Function name
	f_EUTRA_UE_Detach_SwitchOffUe_Trigger

	Reason for change
	Common function is needed to deal with different flavours to trigger the UE to detach or to power off the UE

	Summary of change
	New function 

	TTCN module
	EUTRA_NASSteps


	  type enumerated EUTRA_UE_Detach_SwitchOffUe_UtCmd_Type {powerOff, switchOff, detach};

	  function f_EUTRA_UE_Detach_SwitchOffUe_Trigger(EUTRA_CellId_Type p_CellId,

                                                 template (omit) EUTRA_UE_Detach_SwitchOffUe_UtCmd_Type p_UtCmd := omit,

                                                 boolean p_USIM_Rmvd := false) //@sic R5s140906 sic@

    runs on EUTRA_PTC return EUTRA_UE_Detach_SwitchOffUe_UtCmd_Type

  {

    var EUTRA_UE_Detach_SwitchOffUe_UtCmd_Type v_UtCmd;

    f_IP_ChangeEutraCell(IP, p_CellId, tsc_DRB1);    /* @sic R5s150057 change 4 sic@ 

                                                      * Change IP routing table to allow IMS de-registration 

                                                      * NOTE 1: IMS PDN shall always be mapped to DRB1 

                                                      * NOTE 2: any 2nd PDN (e.g. for emergency call) is mapped to DRB1 too but this shall not matter as IMS de-registration is for PDN1 only

                                                      *         (if there is any issue we would need to hand over e.g. the IP address for PDN1 to be associated to DRB1)

                                                      * NOTE 3: DRB needs to be specified as routing table may be configured for UTRAN or GERAN 

                                                      *         (the DRB can be omitted only if it is known that the routing table is configured for EUTRA already) */

    if (isvalue(p_UtCmd)) {

      v_UtCmd := valueof(p_UtCmd);

    }

    else if (pc_SwitchOnOff or p_USIM_Rmvd and pc_USIM_Removal) { // The UE can be switched off @sic R5s09225 sic@ or USIM removed @sic R5s120597r1 sic@

      v_UtCmd := switchOff;

    }

    else {

      v_UtCmd := powerOff

    }

    select (v_UtCmd) {

      case (powerOff) {

        f_UT_PowerOffUE(UT);

      }

      case (switchOff) {

        f_UT_SwitchOffUE(UT, p_USIM_Rmvd); // @sic R5s100584 sic@

      }

      case (detach) {

        f_UT_InitiateDetach(UT);

      }

      case else {

        FatalError(__FILE__, __LINE__, "");

      }

    }

    return v_UtCmd;

  }


Change 3 
	Function name
	f_EUTRA_UE_Detach_SwitchOffUe_DetachOnly

	Reason for change
	1. Function name is confusing

2. parameter p_RRC_TI is not needed

3. f_IP_ChangeEutraCell shall be called in calling function

4. Different order of parameters than for f_EUTRA_UE_Detach_SwitchOffUe

	Summary of change
	1. function renamed

2. p_RRC_TI removed

3. call of f_IP_ChangeEutraCell removed

4. order of parameters aligned with f_EUTRA_UE_Detach_SwitchOffUe

	Note
	function has been introduced as documented in MCC160 comments to R5s150190

	TTCN module
	EUTRA_NASSteps


Before

	  function f_EUTRA_UE_Detach_SwitchOffUe_DetachOnly(EUTRA_CellId_Type p_CellId,

                                                    RRC_TransactionIdentifier p_RRC_TI := tsc_RRC_TI_Def,

                                                    EUTRA_ATTACH_Type p_ForcedAttach := NORMAL,

                                                    template DetachType p_DetachType := omit,

                                                    integer p_SelectedPlmnId := 1,

                                                    EUTRA_RRC_STATE_Type p_RRCState := RRC_IDLE)runs on EUTRA_PTC

  { // @sic R5s11046: new parameter p_SelectedPlmnId sic@

    var NAS_KsiValue v_KsiValue := f_EUTRA_SecurityKSIasme_Get();

    var template (present) NAS_MSG_Indication_Type v_ExpectedDetachRequest := fl_EUTRA_UE_Detach_SwitchOffUe_ExpectedDetachRequest(p_CellId, p_ForcedAttach, p_DetachType); // @sic R5s150190r1 sic@

    var template (present) NAS_MSG_Indication_Type v_ExpectedServiceRequest := cr_NAS_Indication(tsc_SHT_IntegrityProtected, cr_508_SERVICE_REQUEST(v_KsiValue));  // @sic R5s150190r1 sic@

    f_IP_ChangeEutraCell(IP, p_CellId, tsc_DRB1); // @sic R5s150057 change 4 sic@

    select (p_RRCState) {  // @sic R5s150190 r2 sic@

      case (RRC_IDLE) {

        f_EUTRA_UE_DetachMO_Idle(p_CellId,  // @sic R5s150190r1 sic@

                                           v_ExpectedDetachRequest,

                                           v_ExpectedServiceRequest,

                                           p_SelectedPlmnId);

      }

      case (RRC_CONNECTED) {

        alt {

          [] SRB.receive(car_SRB1_NasPdu_IND(p_CellId, v_ExpectedDetachRequest)) { }

          [] SRB.receive(car_SRB2_NasPdu_IND(p_CellId, v_ExpectedDetachRequest)) { }

        }

      }

    }

  }


After
	  function f_EUTRA_UE_DetachMO(EUTRA_CellId_Type p_CellId ,

                               EUTRA_RRC_STATE_Type p_RRCState := RRC_IDLE,

                               EUTRA_ATTACH_Type p_ForcedAttach := NORMAL,

                               template DetachType p_DetachType := omit,

                               integer p_SelectedPlmnId := 1)runs on EUTRA_PTC

  { // @sic R5s11046: new parameter p_SelectedPlmnId sic@

    var NAS_KsiValue v_KsiValue := f_EUTRA_SecurityKSIasme_Get();

    var template (present) NAS_MSG_Indication_Type v_ExpectedDetachRequest := fl_EUTRA_UE_Detach_SwitchOffUe_ExpectedDetachRequest(p_CellId, p_ForcedAttach, p_DetachType); // @sic R5s150190r1 sic@

    var template (present) NAS_MSG_Indication_Type v_ExpectedServiceRequest := cr_NAS_Indication(tsc_SHT_IntegrityProtected, cr_508_SERVICE_REQUEST(v_KsiValue));  // @sic R5s150190r1 sic@

    select (p_RRCState) {  // @sic R5s150190 r2 sic@

      case (RRC_IDLE) {

        f_EUTRA_UE_DetachMO_Idle(p_CellId,  // @sic R5s150190r1 sic@

                                 v_ExpectedDetachRequest,

                                 v_ExpectedServiceRequest,

                                 p_SelectedPlmnId);

      }

      case (RRC_CONNECTED) {

        alt {

          [] SRB.receive(car_SRB1_NasPdu_IND(p_CellId, v_ExpectedDetachRequest)) { }

          [] SRB.receive(car_SRB2_NasPdu_IND(p_CellId, v_ExpectedDetachRequest)) { }

        }

      }

    }

  }


Further related changes

· fl_TC_9_2_1_1_1a_and9_2_1_1_1b_Body:
f_UT_SwitchOffUE(UT, true) replaced by f_EUTRA_UE_Detach_SwitchOffUe_Trigger(v_EUTRA_CellIdIorL, switchOff)

· fl_TC_9_2_2_1_1_Body, fl_TC_9_2_2_1_8_Body: 
f_UT_SwitchOffUE(UT, true) replaced by f_EUTRA_UE_Detach_SwitchOffUe_Trigger(eutra_CellA, switchOff)

· fl_TC_9_2_2_1_6_Body, fl_TC_9_2_2_1_7_Body, fl_TC_9_2_2_1_8_Body, fl_TC_9_2_2_1_9_Body:
f_UT_InitiateDetach(UT) replaced by f_EUTRA_UE_Detach_SwitchOffUe_Trigger (eutra_CellA, detach)
Change 4 
	Function name
	f_EUTRA_UE_Detach_SwitchOffUe

	Reason for change
	1. New function f_EUTRA_UE_Detach_SwitchOffUe_Trigger to be applied
2. f_EUTRA_UE_DetachMO to be applied

3. parameter p_RRC_TI is not needed

4. Parameter p_SwitchOffDone is not needed anymore (see change 6)

	Summary of change
	1. f_EUTRA_UE_Detach_SwitchOffUe_Trigger used
2. f_EUTRA_UE_DetachMO used

3. p_RRC_TI removed
4. p_ SwitchOffDone removed

	TTCN module
	EUTRA_NASSteps


Before

	  function f_EUTRA_UE_Detach_SwitchOffUe(EUTRA_CellId_Type p_CellId,

                                         EUTRA_RRC_STATE_Type p_RRCState,

                                         RRC_TransactionIdentifier p_RRC_TI := tsc_RRC_TI_Def,
                                         EUTRA_ATTACH_Type p_ForcedAttach := NORMAL,

                                         template DetachType p_DetachType := omit,

                                         integer p_SelectedPlmnId := 1,

                                         boolean p_USIM_Rmvd := false, //@sic R5s140906 sic@

                                         boolean p_SwitchOffDone := false) runs on EUTRA_PTC

  { // @sic R5s11046: new parameter p_SelectedPlmnId sic@

    var NAS_KsiValue v_KsiValue := f_EUTRA_SecurityKSIasme_Get();

    var template (present) NAS_MSG_Indication_Type v_ExpectedDetachRequest := fl_EUTRA_UE_Detach_SwitchOffUe_ExpectedDetachRequest(p_CellId, p_ForcedAttach, p_DetachType); // @sic R5s150190r1 sic@

    var template (present) NAS_MSG_Indication_Type v_ExpectedServiceRequest := cr_NAS_Indication(tsc_SHT_IntegrityProtected, cr_508_SERVICE_REQUEST(v_KsiValue)); // @sic R5s150190r1 sic@

    f_IP_ChangeEutraCell(IP, p_CellId, tsc_DRB1);    /* @sic R5s150057 change 4 sic@ 

                                                      * Change IP routing table to allow IMS de-registration 

                                                      * NOTE 1: IMS PDN shall always be mapped to DRB1 

                                                      * NOTE 2: any 2nd PDN (e.g. for emergency call) is mapped to DRB1 too but this shall not matter as IMS de-registration is for PDN1 only

                                                      *         (if there is any issue we would need to hand over e.g. the IP address for PDN1 to be associated to DRB1)

                                                      * NOTE 3: DRB needs to be specified as routing table may be configured for UTRAN or GERAN 

                                                      *         (the DRB can be omitted only if it is known that the routing table is configured for EUTRA already) */

    if (pc_SwitchOnOff or p_USIM_Rmvd and pc_USIM_Removal) { // The UE can be switched off @sic R5s09225 sic@ or USIM removed @sic R5s120597r1 sic@

      if (not p_SwitchOffDone) { //@sic R5s140906 sic@

        f_UT_SwitchOffUE(UT, p_USIM_Rmvd); // @sic R5s100584 sic@

      }

      select (p_RRCState) {

        case (RRC_IDLE) {

          f_EUTRA_UE_DetachMO_Idle(p_CellId,  // @sic R5s150190r1 sic@

                                   v_ExpectedDetachRequest,

                                   v_ExpectedServiceRequest,

                                   p_SelectedPlmnId);

        }

        case (RRC_CONNECTED) {

          alt {

            [] SRB.receive(car_SRB1_NasPdu_IND(p_CellId, v_ExpectedDetachRequest)) { }

            [] SRB.receive(car_SRB2_NasPdu_IND(p_CellId, v_ExpectedDetachRequest)) { }

          }

        }

      }

        //Now the connection can be released

      f_EUTRA_RRC_ConnectionRelease(p_CellId, p_RRC_TI);

    }

    else { // @sic R5s100584 sic@

      // The power must be removed from the UE

      f_UT_PowerOffUE(UT);

      //Reset all RBs

      f_EUTRA_SS_SRBs_DRBs_Reset(p_CellId);   // @sic R5s110423 sic@

    }

  }


After
	  function f_EUTRA_UE_Detach_SwitchOffUe(EUTRA_CellId_Type p_CellId,

                                         EUTRA_RRC_STATE_Type p_RRCState,

                                         EUTRA_ATTACH_Type p_ForcedAttach := NORMAL,

                                         template DetachType p_DetachType := omit,

                                         integer p_SelectedPlmnId := 1,

                                         boolean p_USIM_Rmvd := false) //@sic R5s140906 sic@

    runs on EUTRA_PTC

  { // @sic R5s11046: new parameter p_SelectedPlmnId sic@

    var EUTRA_UE_Detach_SwitchOffUe_UtCmd_Type v_UtCmd := f_EUTRA_UE_Detach_SwitchOffUe_Trigger(p_CellId, omit, p_USIM_Rmvd);
    select (v_UtCmd) {
      case (poweroff) {
        //Reset all RBs

        f_EUTRA_SS_SRBs_DRBs_Reset(p_CellId);   // @sic R5s110423 sic@

      }

      case else {
        f_EUTRA_UE_DetachMO(p_CellId, p_RRCState, p_ForcedAttach, p_DetachType, p_SelectedPlmnId);
        //Now the connection can be released

        f_EUTRA_RRC_ConnectionRelease(p_CellId);

      }
    }
  }


Further changes: 

tsc_RRC_TI_Def removed from calls of f_EUTRA_UE_Detach_SwitchOffUe
Change 5 
	Function name
	f_EUTRA_Postamble

	Reason for change
	No need for  p_RRC_TI 

	Summary of change
	p_RRC_TI removed

	TTCN module
	EUTRA_CommonProcedures


Before

	  function f_EUTRA_Postamble(EUTRA_CellId_Type p_CellId,

                             EUTRA_POSTAMBLE_STATE_Type p_E_State,

                             EUTRA_ATTACH_Type p_ForcedAttach := NORMAL,

                             RRC_TransactionIdentifier p_RRC_TI := tsc_RRC_TI_Def,

                             integer p_SelectedPlmnId := 1) runs on EUTRA_PTC

  { /* @sic This function has been completely changed in R5-103847 sic@ */

    // @sic R5s11046: new parameter p_SelectedPlmnId sic@

    if (pc_SwitchOnOff) // The UE can actually be switched off

      {

        select (p_E_State) {

          case (E1_IDLE)

          {

            //Expect the UE to detach and release RRC connection

            f_EUTRA_UE_Detach_SwitchOffUe(p_CellId, RRC_IDLE, p_RRC_TI, p_ForcedAttach, omit, p_SelectedPlmnId); // @sic R5s110176, R5s110468 sic@

          }

          case (E2_CONNECTED, E3_TESTMODE)

          {

            //Expect the UE to detach and release RRC connection

            f_EUTRA_UE_Detach_SwitchOffUe(p_CellId, RRC_CONNECTED, p_RRC_TI, p_ForcedAttach, omit, p_SelectedPlmnId); // @sic R5s110176, R5s110468 sic@

          }

          case (E2_T3440) // @sic R5s110007 sic@

          {

            //Expect the UE to release RRC connection

            f_EUTRA_RRC_ConnectionRelease(p_CellId);

            f_Delay (5.0); // @sic R5-110803 sic@

            //Expect the UE to detach and release RRC connection

            f_EUTRA_UE_Detach_SwitchOffUe(p_CellId, RRC_IDLE, p_RRC_TI, p_ForcedAttach, omit, p_SelectedPlmnId); // @sic R5s110176, R5s110468 sic@

          }

          case (E4_DEREGISTERED)

          {

            f_UT_SwitchOffUE (UT, false);

          }

        }

      }

    else

      {// The power must be removed from the UE

        f_UT_PowerOffUE (UT);

      }

    //And now remove the cell(s)

    f_EUTRA_ReleaseAllCells();

  }


After
	  function f_EUTRA_Postamble(EUTRA_CellId_Type p_CellId,

                             EUTRA_POSTAMBLE_STATE_Type p_E_State,

                             EUTRA_ATTACH_Type p_ForcedAttach := NORMAL) runs on EUTRA_PTC

  { /* @sic This function has been completely changed in R5-103847 sic@ */

    // @sic R5s11046: new parameter p_SelectedPlmnId sic@

    var integer p_SelectedPlmnId := 1;

    if (pc_SwitchOnOff) { // The UE can actually be switched off

      select (p_E_State) {

        case (E1_IDLE) {

          //Expect the UE to detach and release RRC connection

          f_EUTRA_UE_Detach_SwitchOffUe(p_CellId, RRC_IDLE, p_ForcedAttach, omit, p_SelectedPlmnId); // @sic R5s110176, R5s110468 sic@

        }

        case (E2_CONNECTED, E3_TESTMODE) {

          //Expect the UE to detach and release RRC connection

            f_EUTRA_UE_Detach_SwitchOffUe(p_CellId, RRC_CONNECTED, p_ForcedAttach, omit, p_SelectedPlmnId); // @sic R5s110176, R5s110468 sic@

        }

        case (E2_T3440) { // @sic R5s110007 sic@

          //Expect the UE to release RRC connection

          f_EUTRA_RRC_ConnectionRelease(p_CellId);

          f_Delay (5.0); // @sic R5-110803 sic@

          //Expect the UE to detach and release RRC connection

          f_EUTRA_UE_Detach_SwitchOffUe(p_CellId, RRC_IDLE, p_ForcedAttach, omit, p_SelectedPlmnId); // @sic R5s110176, R5s110468 sic@

        }

        case (E4_DEREGISTERED) {

          f_UT_SwitchOffUE (UT, false);

        }

      }

    }

    else { // The power must be removed from the UE

      f_UT_PowerOffUE (UT);

    }

    //And now remove the cell(s)

    f_EUTRA_ReleaseAllCells();

  }


Further changes: 
'-' as placeholder for p_RRC_TI removed from calls of f_EUTRA_Postamble in f_TC_6_1_1_4_EUTRA, f_TC_6_1_1_4_EUTRA and f_TC_6_1_2_10_EUTRA
Change 6 
	Function name
	f_EUTRA_Postamble_TunneledDetach

	Reason for change
	1. New function f_EUTRA_UE_Detach_SwitchOffUe_Trigger to be applied to achieve cell change for IMS capable UE

2. parameter p_RRC_TI is not needed

3. f_EUTRA_UE_Detach_SwitchOffUe again checks PICS whether to do power-off only or to expect detach; this can be simplified by using f_EUTRA_UE_DetachMO and f_EUTRA_RRC_ConnectionRelease instead

	Summary of change
	1. Use of f_EUTRA_UE_Detach_SwitchOffUe_Trigger instead of f_UT_SwitchOffUE
2. p_RRC_TI removed

3. f_EUTRA_UE_DetachMO and f_EUTRA_RRC_ConnectionRelease used instead of f_EUTRA_UE_Detach_SwitchOffUe

	TTCN module
	MultiLayer_Procedures_EC2K


Before

	  function f_EUTRA_Postamble_TunneledDetach(EUTRA_CellId_Type p_CellId,

                                            EUTRA_POSTAMBLE_STATE_Type p_E_State,

                                            EUTRA_ATTACH_Type p_ForcedAttach := NORMAL,

                                            RRC_TransactionIdentifier p_RRC_TI := tsc_RRC_TI_Def,

                                            integer p_SelectedPlmnId := 1,

                                            CDMA2000_Type p_CDMA2000Type := type1XRTT) runs on EUTRA_PTC

  {

    //var CDMA2000_Type v_CDMA2000Type;

    if (pc_SwitchOnOff) {   // The UE can actually be switched off

      f_UT_SwitchOffUE (UT, false); //@sic R5s140906 sic@

      select (p_E_State) {

        case (E1_IDLE) {

          // Not Handled yet

          f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "");

        }

        case (E2_CONNECTED, E3_TESTMODE) {

          //Expect the UE to detach and release RRC connection

          if (p_CDMA2000Type == type1XRTT) {

            f_CSFBParametersProc_1xRTT(p_CellId, px_RAND);

            f_CDMA2000_PowerDownDetach_1xRTT(p_CellId);

          } else {  // HRPD

            // Not Handled yet

            f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "");

          }

          f_EUTRA_UE_Detach_SwitchOffUe(p_CellId, RRC_CONNECTED, p_RRC_TI, p_ForcedAttach, omit, p_SelectedPlmnId, false, true); // @sic R5s110176, R5s110468 R5s150081 sic@

        }

        case (E2_T3440) {  // @sic R5s110007 sic@

          // Not Handled yet

          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "");

        }

        case (E4_DEREGISTERED) {

          f_UT_SwitchOffUE(UT, false);

        }

      }

    }

    else { // The power must be removed from the UE

      f_UT_PowerOffUE(UT);

    }

    //And now remove the cell(s)

    f_EUTRA_ReleaseAllCells();

  }


After
	  function f_EUTRA_Postamble_TunneledDetach(EUTRA_CellId_Type p_CellId,

                                            EUTRA_POSTAMBLE_STATE_Type p_E_State,

                                            EUTRA_ATTACH_Type p_ForcedAttach := NORMAL,

                                            integer p_SelectedPlmnId := 1,

                                            CDMA2000_Type p_CDMA2000Type := type1XRTT) runs on EUTRA_PTC

  { 
    if (pc_SwitchOnOff) {   // The UE can actually be switched off

      f_EUTRA_UE_Detach_SwitchOffUe_Trigger(p_CellId, switchOff);  //@sic R5s140906 sic@ 

      select (p_E_State) {

        case (E1_IDLE) {

          // Not Handled yet

          f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "");

        }

        case (E2_CONNECTED, E3_TESTMODE) {

          //Expect the UE to detach and release RRC connection

          if (p_CDMA2000Type == type1XRTT) {

            f_CSFBParametersProc_1xRTT(p_CellId, px_RAND);

            f_CDMA2000_PowerDownDetach_1xRTT(p_CellId);

          } else {  // HRPD

            // Not Handled yet

            f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "");

          }

          f_EUTRA_UE_DetachMO(p_CellId, RRC_CONNECTED, p_ForcedAttach, omit, p_SelectedPlmnId); // @sic R5s110176, R5s110468 R5s150081 sic@

          f_EUTRA_RRC_ConnectionRelease(p_CellId);
        }

        case (E2_T3440) {  // @sic R5s110007 sic@

          // Not Handled yet

          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "");

        }

        case (E4_DEREGISTERED) {

          f_UT_SwitchOffUE(UT, false);

        }

      }

    }

    else { // The power must be removed from the UE

      f_UT_PowerOffUE(UT);

    }

    //And now remove the cell(s)

    f_EUTRA_ReleaseAllCells();

  }


