
Change 1 
	Function name
	f_TC_17_2_4_EUTRA

	Reason for change
	The current mbsfn-Subframe Configuration at step 2 utilises subframe 1 and 3 as mbsfn subframe, in ALL frames. This conflicts with the default SIB schedule, where SIB2 is on subframe1 and SIB13 on subframe3. As a result, the UE will not be able to read SIB2 and SIB13 after step 2.

At step 12/13 when updating the SIB info, extra time shall added to allow the UE read the update, similar as in step 1/2.

	Summary of change
	1. Updated the mbsfn-Subframe Configuration at step 2 (Table 17.2.4.3.3-4), to used subframe 2,6 and 8 rather than 1, 3, and 7

2. Update the frame/subfarme number used in step 4, 6, 8, and 10 to match the updated mbsfn-Subframe Configuration.
3. Added 13sec waiting time after SIB update notification at step 12/13

	TTCN module
	MBMS_MTCH_TestCases.ttcn


Before change

	function f_TC_17_2_4_EUTRA() runs on EUTRA_PTC

  { // Reception of PDCCH DCI format 0 and PHICH in MBSFN subframes

    var GutiParameters_Type v_Guti_Params;

    var B6_Type v_SubFrameAlloc;

    var EUTRA_FDD_TDD_Mode_Type v_EUTRA_FDD_TDD_Mode;

    var SubFrameTiming_Type v_Timing;

    var octetstring v_EncodedPDCPPDU;

    var octetstring v_EncodedRLCPDU;

    var integer v_DL_SubFrameNumber;

    var template (value) MBSFN_SubframeConfigList v_MBSFN_SubframeConfigList;

    timer t_Watchdog := 5.0;

    f_EUTRA_Init (c1);

    f_EUTRA_SetSysinfoCombination_MBMS(eutra_Cell1,c15);

    f_EUTRA_CellConfig_Def (eutra_Cell1);

    f_EUTRA_CellConfig_MBMS_Def (eutra_Cell1); // configured def MCCH data scheduling and MTCH

    f_EUTRA_Preamble(eutra_Cell1, STATE2A_TESTLOOP_ModeA);

    v_Guti_Params := f_EUTRA_CellInfo_GetGuti(eutra_Cell1);

    // Disable AS ciphering on UE side.

    f_EUTRA_AS_CipheringAlgorithm_Set(eea0); // set null AS ciphering algorithm

    //@sic R5s140429 sic@

    f_EUTRA_RbEst_Def_DisableBSR_PHR(eutra_Cell1); // Default RB establishment

    f_EUTRA_CloseUE_TestLoopModeA(eutra_Cell1, cs_UE_TestLoopModeA_LB_Setup_NoScaling); // Bring UE in test loop mode A

    v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD(eutra_Cell1);

    if (v_EUTRA_FDD_TDD_Mode == FDD) {

      v_SubFrameAlloc := '101010'B; //corresponds to subframes 1, 3 and 7

      v_DL_SubFrameNumber := 3;

    } else {

      v_SubFrameAlloc := '000010'B; // corresponds to subframe 9

      v_DL_SubFrameNumber := 9; //@sic R5-144406 sic@

    }

    // step to configure SS RLC/PDCP and MAC in No header manipulation mode on SS Side for DRB

    f_SS_Config1TM_DRB_NoMACHeader(eutra_Cell1, tsc_RbId_DRB1, cs_MAC_TestMode_NoHeaderManipulationDL_UL);

    ///////////////////////////////////////////////////////////////////////////////////

    //      Test Body

    //////////////////////////////////////////////////////////////////////////////////

    f_EUTRA_TestBody_Set(true);

    //@siclog "Steps 1 - 2" siclog@

    // Update and start transmitting modified sys info

    f_EUTRA_CellInfo_SetMBSFN_SubframeConfiguration(eutra_Cell1, {cs_MBSFN_SubframeConfig(n1, 0, v_SubFrameAlloc)});

    f_EUTRA_ModifySysinfo(eutra_Cell1, true, RRC_CONNECTED);

    f_Delay(13.0) ;// 2.5 times of modification period =512rf

    // To stop UL Grant Transmision

    f_EUTRA_StopULGrantTransmission_PUCCH_Sync(eutra_Cell1, cs_TimingInfo_Now);

    f_SS_ConfigPhich_TestMode(eutra_Cell1, cs_Phich_ExplicitMode(cs_HARQ_ModeList_2Nack_Ack));

    //@siclog "Step 3" siclog@

    // Send DL MAC PDU

    v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1);

    // 1 MAC PDU = 1 RLC PDU;  1 RLC PDU =1 PDCP PDU

    // MAC Header = 1B

    // 1 MAC SDU of 14 => RLC SDU= 12, PDCP SDU = 10

    // Total MAC PDU = 1 + 14  = 15B = 120b (best fit)

    v_EncodedPDCPPDU := f_PDCP_PDU_Encvalue(cs_PDCP_PDU_LongSN(cs_PDCP_DataPdu_LongSN(tsc_B12_SQN0, crs_PDCP_SDU_10B)));

    v_EncodedRLCPDU  := f_RLC_PDU_Encvalue(cs_RLC_AMD_1PDU(tsc_NoPoll, int2bit(0, 10), v_EncodedPDCPPDU));

    DRB.send (cas_DRB_COMMOM_REQ_MAC_PDU(eutra_Cell1,

                                         tsc_RbId_DRB1,

                                         cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                         cs_MAC_PDUList_1 (cs_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1, v_EncodedRLCPDU))));

    //@siclog "Step 4" siclog@

    // UL Grant

    v_Timing.SFN.Number := (v_Timing.SFN.Number + 6) mod 1024;

    v_Timing.Subframe.Number := v_DL_SubFrameNumber; // 3 for FDD, 9 for TDD

    f_EUTRA_OneULGrantTransmission( eutra_Cell1,

                                    cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                    4, /* Nprb = 2, */

                                    2); /* Itbs = 4,   Table 7.1.7.2.1-1 of 36.213: Imcs = 4 -> Itbs = 4   MCS index of table 8.6.1-1 of 36.213=> 120 bits */

    //@siclog "Step 5" siclog@

    // UL MAC PDU

    v_Timing := f_GetSubFrameNumForULData(v_EUTRA_FDD_TDD_Mode, v_Timing);

    t_Watchdog.start;

    DRB.receive (car_DRB_COMMON_IND_MAC(eutra_Cell1,

                                        tsc_RbId_DRB1,

                                        cr_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                        {cr_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1, cr_Octet14_Any)}));

    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Step 5");

    v_Timing := f_EvaluateNext_UL_TransmissionForHARQ_Process(v_Timing, v_EUTRA_FDD_TDD_Mode); // Next TTI.

    //@siclog "Step 6" siclog@

    // Automatically taken care by SS

    //@siclog "Step 7" siclog@

    // UL MAC PDU

    DRB.receive (car_DRB_COMMON_IND_MAC(eutra_Cell1,

                                        tsc_RbId_DRB1,

                                        cr_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                        {cr_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1, cr_Octet14_Any)}));

    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Step 7");

    v_Timing := f_EvaluateNext_UL_TransmissionForHARQ_Process(v_Timing, v_EUTRA_FDD_TDD_Mode); // Next TTI.

    //@siclog "Step 8" siclog@

    // Automatically taken care by SS

    //@siclog "Step 9" siclog@

    // UL MAC PDU

    DRB.receive (car_DRB_COMMON_IND_MAC(eutra_Cell1,

                                        tsc_RbId_DRB1,

                                        cr_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                        {cr_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1, cr_Octet14_Any)}));

    //@siclog "Step 10" siclog@

    // Automatically taken care by SS

    f_RLC_Status_PDU_Tx (tsc_LchId_DRB1, 1); //@sic R5s140585 sic@

    //@siclog "Step 11" siclog@

    alt {

      [] DRB.receive (car_DRB_COMMON_IND_MAC(eutra_Cell1,

                                             tsc_RbId_DRB1,

                                             cr_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                             {cr_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1, cr_Octet14_Any)}))

        {

          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 11: UE retransmits MAC PDU");

          t_Watchdog.stop;

        }

      []t_Watchdog.timeout

        {

          f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Step 11: UE does not retransmits MAC PDU");

        }

    }

    // configure SS for default PHICH mode

    f_SS_ConfigPhich_TestMode (eutra_Cell1, cs_PhichTestModeNormal);

    f_SS_Config1TM_DRB_NoMACHeader(eutra_Cell1, tsc_RbId_DRB1, cs_MAC_TestMode_NormalMode); // configure SS in normal mode

    // configure SS for default grant allocation

    f_EUTRA_StartDefULGrantTransmission(eutra_Cell1);

    //@siclog "Steps 12 - 13" siclog@

    // Update and start transmitting modified sys info

    if (v_EUTRA_FDD_TDD_Mode == FDD) {

      v_MBSFN_SubframeConfigList := {

        cs_MBSFN_SubframeConfig_Def(1, '100000'B)

      };

    } else {

      v_MBSFN_SubframeConfigList := {

        cs_MBSFN_SubframeConfig_Def(0, '000010'B)

      };

    }

    f_EUTRA_CellInfo_SetMBSFN_SubframeConfiguration(eutra_Cell1,v_MBSFN_SubframeConfigList);

    f_EUTRA_ModifySysinfo(eutra_Cell1, true, RRC_CONNECTED);

    f_EUTRA_TestBody_Set(false);

    //test body ends

    ///////////////////////////////////////////////////////////////////////////////////

    //      Post Amble

    //////////////////////////////////////////////////////////////////////////////////

    // Release RRC Connection and Switch off

    f_EUTRA_Postamble(eutra_Cell1, E3_TESTMODE);

  }

  // End of  module

}


After change
	function f_TC_17_2_4_EUTRA() runs on EUTRA_PTC

  { // Reception of PDCCH DCI format 0 and PHICH in MBSFN subframes

    var GutiParameters_Type v_Guti_Params;

    var B6_Type v_SubFrameAlloc;

    var EUTRA_FDD_TDD_Mode_Type v_EUTRA_FDD_TDD_Mode;

    var SubFrameTiming_Type v_Timing;

    var octetstring v_EncodedPDCPPDU;

    var octetstring v_EncodedRLCPDU;

    var integer v_DL_SubFrameNumber;

    var template (value) MBSFN_SubframeConfigList v_MBSFN_SubframeConfigList;

    timer t_Watchdog := 5.0;

    f_EUTRA_Init (c1);

    f_EUTRA_SetSysinfoCombination_MBMS(eutra_Cell1,c15);

    f_EUTRA_CellConfig_Def (eutra_Cell1);

    f_EUTRA_CellConfig_MBMS_Def (eutra_Cell1); // configured def MCCH data scheduling and MTCH

    f_EUTRA_Preamble(eutra_Cell1, STATE2A_TESTLOOP_ModeA);

    v_Guti_Params := f_EUTRA_CellInfo_GetGuti(eutra_Cell1);

    // Disable AS ciphering on UE side.

    f_EUTRA_AS_CipheringAlgorithm_Set(eea0); // set null AS ciphering algorithm

    //@sic R5s140429 sic@

    f_EUTRA_RbEst_Def_DisableBSR_PHR(eutra_Cell1); // Default RB establishment

    f_EUTRA_CloseUE_TestLoopModeA(eutra_Cell1, cs_UE_TestLoopModeA_LB_Setup_NoScaling); // Bring UE in test loop mode A

    v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD(eutra_Cell1);

    if (v_EUTRA_FDD_TDD_Mode == FDD) {

      v_SubFrameAlloc := '010101'B; //corresponds to subframes 2, 6 and 8 //CHANGE 1
      v_DL_SubFrameNumber := 8; //CHANGE 2
    } else {

      v_SubFrameAlloc := '000010'B; // corresponds to subframe 9

      v_DL_SubFrameNumber := 9; //@sic R5-144406 sic@

    }

    // step to configure SS RLC/PDCP and MAC in No header manipulation mode on SS Side for DRB

    f_SS_Config1TM_DRB_NoMACHeader(eutra_Cell1, tsc_RbId_DRB1, cs_MAC_TestMode_NoHeaderManipulationDL_UL);

    ///////////////////////////////////////////////////////////////////////////////////

    //      Test Body

    //////////////////////////////////////////////////////////////////////////////////

    f_EUTRA_TestBody_Set(true);

    //@siclog "Steps 1 - 2" siclog@

    // Update and start transmitting modified sys info

    f_EUTRA_CellInfo_SetMBSFN_SubframeConfiguration(eutra_Cell1, {cs_MBSFN_SubframeConfig(n1, 0, v_SubFrameAlloc)});

    f_EUTRA_ModifySysinfo(eutra_Cell1, true, RRC_CONNECTED);

    f_Delay(13.0) ;// 2.5 times of modification period =512rf

    // To stop UL Grant Transmision

    f_EUTRA_StopULGrantTransmission_PUCCH_Sync(eutra_Cell1, cs_TimingInfo_Now);

    f_SS_ConfigPhich_TestMode(eutra_Cell1, cs_Phich_ExplicitMode(cs_HARQ_ModeList_2Nack_Ack));

    //@siclog "Step 3" siclog@

    // Send DL MAC PDU

    v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1);

    // 1 MAC PDU = 1 RLC PDU;  1 RLC PDU =1 PDCP PDU

    // MAC Header = 1B

    // 1 MAC SDU of 14 => RLC SDU= 12, PDCP SDU = 10

    // Total MAC PDU = 1 + 14  = 15B = 120b (best fit)

    v_EncodedPDCPPDU := f_PDCP_PDU_Encvalue(cs_PDCP_PDU_LongSN(cs_PDCP_DataPdu_LongSN(tsc_B12_SQN0, crs_PDCP_SDU_10B)));

    v_EncodedRLCPDU  := f_RLC_PDU_Encvalue(cs_RLC_AMD_1PDU(tsc_NoPoll, int2bit(0, 10), v_EncodedPDCPPDU));

    DRB.send (cas_DRB_COMMOM_REQ_MAC_PDU(eutra_Cell1,

                                         tsc_RbId_DRB1,

                                         cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                         cs_MAC_PDUList_1 (cs_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1, v_EncodedRLCPDU))));

    //@siclog "Step 4" siclog@

    // UL Grant

    v_Timing.SFN.Number := (v_Timing.SFN.Number + 6) mod 1024;

    v_Timing.Subframe.Number := v_DL_SubFrameNumber; // 3 for FDD, 9 for TDD

    f_EUTRA_OneULGrantTransmission( eutra_Cell1,

                                    cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                    4, /* Nprb = 2, */

                                    2); /* Itbs = 4,   Table 7.1.7.2.1-1 of 36.213: Imcs = 4 -> Itbs = 4   MCS index of table 8.6.1-1 of 36.213=> 120 bits */

    //@siclog "Step 5" siclog@

    // UL MAC PDU

    v_Timing := f_GetSubFrameNumForULData(v_EUTRA_FDD_TDD_Mode, v_Timing);

    t_Watchdog.start;

    DRB.receive (car_DRB_COMMON_IND_MAC(eutra_Cell1,

                                        tsc_RbId_DRB1,

                                        cr_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                        {cr_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1, cr_Octet14_Any)}));

    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Step 5");

    v_Timing := f_EvaluateNext_UL_TransmissionForHARQ_Process(v_Timing, v_EUTRA_FDD_TDD_Mode); // Next TTI.

    //@siclog "Step 6" siclog@

    // Automatically taken care by SS

    //@siclog "Step 7" siclog@

    // UL MAC PDU

    DRB.receive (car_DRB_COMMON_IND_MAC(eutra_Cell1,

                                        tsc_RbId_DRB1,

                                        cr_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                        {cr_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1, cr_Octet14_Any)}));

    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Step 7");

    v_Timing := f_EvaluateNext_UL_TransmissionForHARQ_Process(v_Timing, v_EUTRA_FDD_TDD_Mode); // Next TTI.

    //@siclog "Step 8" siclog@

    // Automatically taken care by SS

    //@siclog "Step 9" siclog@

    // UL MAC PDU

    DRB.receive (car_DRB_COMMON_IND_MAC(eutra_Cell1,

                                        tsc_RbId_DRB1,

                                        cr_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                        {cr_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1, cr_Octet14_Any)}));

    //@siclog "Step 10" siclog@

    // Automatically taken care by SS

    f_RLC_Status_PDU_Tx (tsc_LchId_DRB1, 1); //@sic R5s140585 sic@

    //@siclog "Step 11" siclog@

    alt {

      [] DRB.receive (car_DRB_COMMON_IND_MAC(eutra_Cell1,

                                             tsc_RbId_DRB1,

                                             cr_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                             {cr_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1, cr_Octet14_Any)}))

        {

          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 11: UE retransmits MAC PDU");

          t_Watchdog.stop;

        }

      []t_Watchdog.timeout

        {

          f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Step 11: UE does not retransmits MAC PDU");

        }

    }

    // configure SS for default PHICH mode

    f_SS_ConfigPhich_TestMode (eutra_Cell1, cs_PhichTestModeNormal);

    f_SS_Config1TM_DRB_NoMACHeader(eutra_Cell1, tsc_RbId_DRB1, cs_MAC_TestMode_NormalMode); // configure SS in normal mode

    // configure SS for default grant allocation

    f_EUTRA_StartDefULGrantTransmission(eutra_Cell1);

    //@siclog "Steps 12 - 13" siclog@

    // Update and start transmitting modified sys info

    if (v_EUTRA_FDD_TDD_Mode == FDD) {

      v_MBSFN_SubframeConfigList := {

        cs_MBSFN_SubframeConfig_Def(1, '100000'B)

      };

    } else {

      v_MBSFN_SubframeConfigList := {

        cs_MBSFN_SubframeConfig_Def(0, '000010'B)

      };

    }

    f_EUTRA_CellInfo_SetMBSFN_SubframeConfiguration(eutra_Cell1,v_MBSFN_SubframeConfigList);

    f_EUTRA_ModifySysinfo(eutra_Cell1, true, RRC_CONNECTED);

    f_Delay(13.0) ;// 2.5 times of modification period =512rf //CHANGE 3
    f_EUTRA_TestBody_Set(false);

    //test body ends

    ///////////////////////////////////////////////////////////////////////////////////

    //      Post Amble

    //////////////////////////////////////////////////////////////////////////////////

    // Release RRC Connection and Switch off

    f_EUTRA_Postamble(eutra_Cell1, E3_TESTMODE);

  }

  // End of  module

}


